
Paper SA15-2014 
 

SAS®: Data Manipulation Tools 
Audrey Yeo, Athene USA, West Des Moines, Iowa 

 
Abstract 
Let’s face it, the data provided to us is typically never easy to work with. Missing values, dates that aren’t populated 
the way we need (e.g. Year, Month, Day vs. Month, Day, Year), leading zeros that need to be removed from a string, 
leading zeros that need to be added to a string, to name a few. We must work through and massage the data before 
we can start creating reports with it. In this paper, we present a collection of basic data manipulation tools that should 
help make the data cleaning easier. 

1. PROC STDIZE – Can be used to replace missing values 
 
It is easy to change missing values to a zero from a . (dot) in a DATA step, as shown below  

IF MISSING(VAR) THEN VAR = 0; 
 
However, PROC STDIZE is a much simpler and effective tool to do that. 
 
Data set (Input): 
 

 
 
Code: 
 
PROC STDIZE DATA=CARS REPONLY MISSING=0 OUT=CARS2; 
 VAR MSRP;  
RUN; 
 
 
 
 
 



Data set (Output): 
 

 

Let’s say we have the input data set above, where there are some missing values in the MSRP column, and we want 
to replace all missing values in that column with the value 0.  

The REPONLY option in the code above tells SAS to replace only missing data and the MISSING=0 option tells SAS 
that we want the missing value to be replaced with a value 0. 

Let’s say we have multiple columns with missing numeric values (as shown below) and we want to substitute all the 
numeric missing values with 0s. 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 



Data set (Input): 
 

 

Code: 
 
PROC STDIZE DATA=CARS3 REPONLY MISSING=0 OUT=CARS4;  
 VAR _NUMERIC_; 
RUN; 

Data set (Output): 
 

 



We can do that by changing the specific variable name from MSRP to _NUMERIC_, this tells SAS that we want all 
numeric columns with missing values be substituted with the value 0. 

We can also tell SAS to change the missing values to values other than 0s. 

Code: 
 
PROC STDIZE DATA=CARS3 REPONLY MISSING=-1 OUT=CARS5;  
 VAR _NUMERIC_; 
RUN; 

Data set (Output): 
 

 

All we need to do is to change the MISSING=0 option to the value that we want. In the example above, we changed 
all the missing values to a value of -1. 

2. INPUT function – Use to convert character variables into numeric variables 
 

For this example, we have information in the form of character which we want to be numeric information. This can be 
done using the INPUT function with an informat.  

Code: 
 
DATA NUMERIC; 
 NUM_STRING1 = '12345'; 
 NUM_STRING2 = '123'; 
 
 NUM1 = INPUT(NUM_STRING1,8.); 
 NUM2 = INPUT(NUM_STRING2,8.); 
RUN; 
 



Data set (Output): 
 

 
 
In the example above, we have two character strings, num_string1 and num_string2. We use the INPUT function with 
an 8. informat to tell SAS to “read” in num_string1 and convert it to a numeric string and assign that value to a new 
variable num1.  We did the same thing to convert num_String2 to num2. The numeric informat of 8 tells SAS how to 
read in the form of the values that needs to be converted. 
 
In addition to numeric string, we’ve also included an example on using the INPUT function to convert a character 
string of numbers to a date. 

Code: 
 
DATA DATES; 
 DATE_STRING1 = '20140508'; 
 DATE_STRING2 = '03082014'; 
 DATE_STRING3 = '21082014'; 
 
 DATE_NUMERIC1 = INPUT(DATE_STRING1,YYMMDD10.); 
 DATE1 = INPUT(DATE_STRING1,YYMMDD10.); 
 FORMAT DATE1 YYMMDD10.; 
 
 DATE_NUMERIC2 = INPUT(DATE_STRING2,MMDDYY10.); 
 DATE2 = INPUT(DATE_STRING2,MMDDYY10.); 
 FORMAT DATE2 MMDDYY10.; 
 
 DATE_NUMERIC3 = INPUT(DATE_STRING3,DDMMYY10.); 
 DATE3 = INPUT(DATE_STRING3,DDMMYY10.); 
 FORMAT DATE3 DDMMYY10.; 
RUN; 
 
Data set (Output): 
 

 
 
We have three different date strings in the data set above, date_string1, date_string2, and date_string3 respectively. 
Date_string1 is in the year, month, day informat; date_string2 is in the month, day, year informat while date_string3 is 
in the day, month, year informat.  
 
Next, we used the INPUT function with the respective date informats to change the character information into numeric 
values, as shown in the output data set above, date_numeric1, date_numeric2, and date_numeric3, respectively. To 
make it easier to read the dates, we add a date format with the FORMAT statement and results can be seen in date1, 
date2, and date3. 

3. PUT function – Use to convert numeric variables into character variables 

Now that we know how to change character variables into numeric variables, it only makes sense to demonstrate 
how to convert character variables into numeric variables. In the example below, the variable “code” is in numeric 
format, and we wish to add leading zeros to it if the variable has less than five characters.  

 
 
 



Code: 
 
DATA NUMBERS; 
 INPUT CODE 5.; 
 DATALINES; 
34639 
34234 
93473 
 5495 
23748 
 4857 
  548 
 4589 
   45 
  459 
  345 
43555 
 3458 
; 
RUN; 
 
DATA NUMBERS2; 
 SET NUMBERS; 
 CODE2 = PUT(CODE, z5.); 
RUN; 
 
Data set (Output): 

 

We have a list of code in numeric format in the input data set above that we would like to add leading zeros in front of 
it if the value has less than five characters. This can be done using the PUT function with a format. The format (2nd 
argument of the PUT function) tells SAS how we want the output to be presented. 

As can be seen in the data set output above, we have the original “code” column in numeric format, after using the 
PUT function with a z5. format, we’re able to convert the numeric column into a character string with leading zeros for 
variables that has less than five characters.  

In addition, we can also use PUT function to convert a numerical value to a date character. 

 



Code: 

DATA DATES_C; 
 
 DATE1 = 19758; 
 DATE2 = 1; 
 
 CHAR_DATE1 = PUT(DATE1,DATE9.); 
 CHAR_DATE2 = PUT(DATE2,DATE9.); 
RUN; 

Data set (Output): 

 

In the code above, we have DATE1 and DATE2 in numeric format and we want to convert it into a date character 
string. Using a DATE9. format as the 2nd argument of the PUT function tells SAS the format that we want DATE1 and 
DATE2 to be formatted and assign them to CHAR_DATE1 and CHAR_DATE2 respectively. 

4. LAG function – Can be used to access a previous numeric value 

We have a list of S&P Indexes collected daily and we want to compare today’s index value to the previous day’s 
index value. We can use the LAG function to do this.  

Code: 

DATA SP500MAP1; 
  INPUT INDEXDATE DDMMYY10. SP500; 
  DATALINES;  
01/01/2008  1468.4 
02/01/2008  1447.2 
03/01/2008  1447.2 
04/01/2008  1411.6 
05/01/2008  1411.6 
06/01/2008  1411.6 
07/01/2008  1416.2 
08/01/2008  1390.2 
09/01/2008  1409.1 
10/01/2008  1420.3 
11/01/2008  1401.0 
;  
RUN;  
 
DATA SP500MAP2; 
 SET SP500MAP1; 
 SP500_PREVIOUS = LAG(SP500); 
RUN; 
 
 

 

 



Data set (Output): 

 
 
In order to compare today’s index values to the previous day’s index value, we created a new column named 
sp500_previous, using the LAGn function. The LAG function in the example code can also be written as LAG1, which 
tells SAS to return one missing value and the values of sp500 lagged once. 
 
Let’s say we want to compare the index values five days apart, or even ten days apart, we can do this by changing 
LAG1 to LAG5 (for five days apart) or LAG10 (for ten days apart), as shown in the examples below. 
 
Code: 

DATA SP500MAP3; 
 SET SP500MAP1; 
 SP500_PREVIOUS = LAG5(SP500); 
RUN; 
 
Data set (Output): 

 

LAG5 in the example above tells SAS to return five missing values and the values of sp500 lagged five times. 

Code: 

DATA SP500MAP4; 
 SET SP500MAP1; 
 FORMAT INDEXDATE MMDDYY10.; 
 SP500_PREVIOUS = LAG10(SP500); 
RUN; 

 



Data set (Output): 

 

LAG10 in the example above tells SAS to return ten missing values and the values of sp500 lagged ten times. 

5. SUBSTR function – Use to parse a text string 

Let’s say we have an ID column that can be split into two different columns. We can use the SUBSTR function to do 
that. 

Code: 

DATA TEXT_SPLIT; 
 INPUT ID $9.; 
 DATALINES; 
234367ABC 
314631XYZ 
347834PDQ 
348324ABC 
343267PDQ 
346327PDQ 
234673AIL 
734627XYZ 
482137XYZ 
346328XYZ 
; 
RUN; 
 
DATA TEXT_SPLIT2; 
 SET TEXT_SPLIT; 
 ID_NO = SUBSTR(ID,1,6); 
 ID_CODE = SUBSTR(ID,7,3); 
RUN; 

 

 

 



Data set (Output): 

 

In this example, we have a column called ID that can be split into two different columns, ID_no and ID_code. The first 
6 characters of the ID column is the ID_no and the remaining 3 characters make up the ID_code. Here, we use two 
SUBSTR functions to split the ID up. 

In the first SUBSTR function, we named the new column as ID_no, and tell SAS to pull the first character from the ID 
column and we want up to the sixth character of the ID column. 

In the second SUBSTR function, we tell SAS to pull the seventh character from the ID column and up to three 
characters and assign it to the ID_Code column. 

6. SCAN  function – Searches a string for a defined substring of text 

The SUBSTR function is a very useful function if we want to parse a text string. The SCAN function is another useful 
function for parsing text, especially if we have delimiters in it. 

Code: 

DATA ALL; 
 INPUT VALGROUP $35.; 
 DATALINES; 
ABC 10-High 
ABC 10-Low 
ABC 6-High 
ABC 6-Low 
ABC Bonus Pro-High 
ABC Bonus Pro-Low 
ABC Bonus-High 
ABC Bonus-Low 
ABC Ultra-High 
ABC Ultra-Low 
XYZ 10-High 
XYZ 10-Low 
XYZ 5-High 
XYZ 5-Low 
XYZ 7-High 
XYZ 7-Low 
;  
RUN;  



DATA PROD_BAND; 
 SET ALL; 
 PRODUCT = SCAN(VALGROUP,1,"-"); 
 BAND = SCAN(VALGROUP,-1,"-"); 
run; 

Data set (Output): 

 

In this example above, we have a column called valgroup which consists of the product name and whether it is a high 
band product or a low band product. Using the SCAN function, we are able to parse the information into two different 
types of information.  

In the first SCAN function, we are asking SAS to extract the first word, as indicated by the 2nd argument of the 
function, from the variable valgroup until we see a ‘-‘ and assign that word to the column product.  

Using a negative value in the 2nd argument of the function tells SAS to proceed to scan from right to left and extract 
the first word until we see a ‘-‘ and assign that word to the column band. 

7. UPCASE, LOWCASE, PROPCASE function – Changes the case of the text variables 
into uppercase, lowercase and proper case respectively 

Sometimes we get information that has inconsistent cases, especially if it is inputted manually. We can always 
standardize the variable by changing it to all uppercase, lower case, or proper case.  

 
 
 
 
 
 
 
 
 



Code: 

DATA CASE; 
 INPUT STATE $15.; 
 UPPER = UPCASE(STATE); 
 LOWER = LOWCASE(STATE); 
 PROPER = PROPCASE(STATE); 
 DATALINES; 
NeW York 
cAlifornIA 
CHicago 
SEATTLE 
; 
RUN; 

Data set (Output): 

 

The ability to standardize the case of a variable is very useful. It is difficult to subset or filter a variable if they differ by 
case. Using the UPCASE function, we are able to change all the text into upper case text; LOWCASE function 
changes the text into lower case test; and PROPCASE function changes the text into their appropriate cases. 

Summary: 

Data cleaning can be time consuming and frustrating. Having a few tricks up your sleeve could really expedite the 
process and make things easier. 

